Issue XXXIX: Topos on Category of Sets

Maxim Sokhatsky

[1] National Technical University of Ukraine
Igor Sikorsky Kyiv Polytechnical Institute
15 травня 2025 р.

Анотація

The purpose of this work is to clarify all topos definitions using type theory. Not much efforts was done to give all the examples, but one example, a topos on category of sets, is constructively presented at the finale.

As this cricial example definition is used in presheaf definition, the construction of category of sets is a mandatory excercise for any topos library. We propose here cubicaltt[1] version of elementary topos on category of sets for demonstration of categorical semantics (from logic perspective) of the fundamental notion of set theory in mathematics.

Other disputed foundations for set theory could be taken as: ZFC, NBG, ETCS. We will disctinct syntetically: i) category theory; ii) set theory in univalent foundations; iii) topos theory, grothendieck topos, elementary topos. For formulation of definitions and theorems only Martin-Löf Type Theory is requested. The proofs involve cubical type checker primitives.

Keywords: Homotopy Type Theory, Topos Theory

Зміст

[1] Cubical Type Theory, http://github.com/mortberg/cubicaltt

1 Topos Theory

One can admit two topos theory lineages. One lineage takes its roots from published by Jean Leray in 1945 initial work on sheaves and spectral sequences. Later this lineage was developed by Henri Paul Cartan, André Weil. The peak of lineage was settled with works by Jean-Pierre Serre, Alexander Grothendieck, and Roger Godement.

Second remarkable lineage take its root from William Lawvere and Myles Tierney. The main contribution is the reformulation of Grothendieck topology by using subobject classifier.

1.1 Set Theory

Here is given the ∞-groupoid model of sets.

Definition 1. (Mere proposition, PROP). A type P is a mere proposition if for all $x, y : P$ we have $x = y$:

$$\mathrm{isProp}(P) = \prod_{x,y:P} (x = y).$$

Definition 2. (0-type). A type A is a 0-type is for all $x, y : A$ and $p, q : x =_A y$ we have $p = q$.

Definition 3. (1-type). A type A is a 1-type if for all $x, y : A$ and $p, q : x =_A y$ and $r, s : p =_{=_A} q$, we have $r = s$.

Definition 4. (A set of elements, SET). A type A is a SET if for all $x, y : A$ and $p, q : x = y$, we have $p = q$:

$$\mathrm{isSet}(A) = \prod_{x,y:A} \prod_{p,q:x=y} (p = q).$$

Definition 5.

```
data N = Z  | S (n: N)

n_grpd (A: U) (n: N): U = (a b: A) -> rec A a b n where
  rec (A: U) (a b: A) : (k: N) -> U
    = split { Z -> Path A a b ; S n -> n_grpd (Path A a b) n }

isContr (A: U): U = (x: A) * ((y: A) -> Path A x y)
isProp  (A: U): U = n_grpd A Z
isSet   (A: U): U = n_grpd A (S Z)
PROP  : U = (X:U) * isProp X
SET   : U = (X:U) * isSet X
```

Definition 6. (Π-Contractability). If fiber is set thene path space between any sections is contractible.

```
setPi (A: U) (B: A -> U) (h: (x: A) -> isSet (B x)) (f g: Pi A B)
      (p q: Path (Pi A B) f g)
    : Path (Path (Pi A B) f g) p q
```

Definition 7. (Σ-Contractability). If fiber is set then Σ is set.

```
setSig (A:U) (B: A -> U) (base: isSet A)
       (fiber: (x:A) -> isSet (B x)) : isSet (Sigma A B)
```

Definition 8. (Unit type, $\mathbb{1}$). The unit $\mathbb{1}$ is a type with one element.

```
data unit = tt
unitRec (C: U) (x: C): unit -> C = split tt -> x
unitInd (C: unit -> U) (x: C tt): (z:unit) -> C z
  = split tt -> x
```

Theorem 1. (Category of Sets, $\mathbf{Set}$). Sets forms a Category. All compositional theorems proved by using reflection rule of internal language. The proof that Hom forms a set is taken through Π-contractability.

```
Set: precategory = ((Ob,Hom),id,c,HomSet,L,R,Q) where
  Ob: U = SET
  Hom (A B: Ob): U = A.1 -> B.1
  id (A: Ob): Hom A A = idfun A.1
  c (A B C: Ob) (f: Hom A B) (g: Hom B C): Hom A C
    = o A.1 B.1 C.1 g f
  HomSet (A B: Ob): isSet (Hom A B) = setFun A.1 B.1 B.2
  L (A B:Ob) (f:Hom A B): Path (Hom A B)(c A A B (id A)f)f
    = refl (Hom A B) f
  R (A B:Ob) (f:Hom A B): Path (Hom A B)(c A B B f(id B))f
    = refl (Hom A B) f
  Q (A B C D: Ob) (f:Hom A B) (g:Hom B C) (h:Hom C D)
    : Path (Hom A D) (c A C D (c A B C f g) h)
                     (c A B D f (c B C D g h))
    = refl (Hom A D) (c A B D f (c B C D g h))
```

1.2 Topological Structure

Topos theory extends category theory with notion of topological structure but reformulated in a categorical way as a category of sheaves on a site or as one that has cartesian closure and subobject classifier. We give here two definitions.

Definition 9. (Topology). The topological structure on A (or topology) is a subset $S \in A$ with following properties: i) any finite union of subsets of S is belong to S; ii) any finite intersection of subsets of S is belong to S. Subets of S are called open sets of family S.

```
def =₁ (A : U₁) (x y : A) := PathP (<_> A) x y
def isProp₁ (A : U₁) := Π (a b : A), =₁ A a b
def isSet₁ (A : U₁) := Π (a b : A) (x y : =₁ A a b), =₁
(=₁ A a b) x y
def Prop := U → 2
def ℙ (X: U₁) := X → Prop

def ∅ (X: U₁) : ℙ X
 := λ (_: X) (_: U), false

def total (X: U₁) : ℙ X
 := λ (_: X) (_: U), true

def ∈ (X: U₁) (el: X) (set: ℙ X) : U₁
 := =₁ (U → 2) (set el) (\(_: U), true)

def ∉ (X: U₁) (el: X) (set: ℙ X) : U₁
 := =₁ (U → 2) (set el) (\(_: U), false)

def ⊆ (X: U₁) (A B: ℙ X)
 := Π (x: X), (∈ X x A) × (∈ X x B)

def ⊆ (X: U₁) : ℙ X → ℙ X
 := λ (h  : ℙ X), λ (x: X) (Y: U), not (h x Y)

def ∪ (X: U₁) : ℙ X → ℙ X → ℙ X
 := λ (h1 : ℙ X) (h2: ℙ X), λ (x: X) (Y: U), or (h1 x Y) (h2 x Y)

def ∩ (X: U₁) : ℙ X → ℙ X → ℙ X
 := λ (h1 : ℙ X) (h2: ℙ X), λ (x: X) (Y: U), and (h1 x Y) (h2 x Y)
```

For fully functional general topology theorems and Zorn lemma you can refer to the Coq library [2]topology by Daniel Schepler.

[2]https://github.com/verimath/topology

1.3 Grothendieck Topos

Grothendieck Topology is a calculus of coverings which generalizes the algebra of open covers of a topological space, and can exist on much more general categories. There are three variants of Grothendieck topology definition: i) sieves; ii) coverage; iii) covering families. A category have one of these three is called a Grothendieck site.

Examples: Zariski, flat, étale, Nisnevich topologies.

A sheaf is a presheaf (functor from opposite category to category of sets) which satisties patching conditions arising from Grothendieck topology, and applying the associated sheaf functor to preashef forces compliance with these conditions.

The notion of Grothendieck topos is a geometric flavour of topos theory, where topos is defined as category of sheaves on a Grothendieck site with geometric moriphisms as adjoint pairs of functors between topoi, that satisfy exactness properties. [**?**]

As this flavour of topos theory uses category of sets as a prerequisite, the formal construction of set topos is cricual in doing sheaf topos theory.

Definition 10. (Sieves). Sieves are a family of subfunctors

$$R \subset \mathrm{Hom}_C(\_, U), U \in C,$$

such that following axioms hold: i) (base change) If $R \subset \mathrm{Hom}_C(\_, U)$ is covering and $\phi : V \to U$ is a morphism of C, then the subfuntor

$$\phi^{-1}(R) = \{\gamma : W \to V \| \phi \cdot \gamma \in R\}$$

is covering for V; ii) (local character) Suppose that $R, R' \subset \mathrm{Hom}_C(\_, U)$ are subfunctors and R is covering. If $\phi^{-1}(R')$ is covering for all $\phi : V \to U$ in R, then R' is covering; iii) $\mathrm{Hom}_C(\_, U)$ is covering for all $U \in C$.

Definition 11. (Coverage). A coverage is a function assigning to each Ob_C the family of morphisms $\{f_i : U_i \to U\}_{i\in I}$ called covering families, such that for any $g : V \to U$ exist a covering family $\{h : V_j \to V\}_{j\in J}$ such that each composite $h_j \circ g$ factors some f_i:

$$\begin{array}{ccc} V_j & \xrightarrow{k} & U_i \\ \downarrow{\scriptstyle h} & & \downarrow{\scriptstyle f_i} \\ V & \xrightarrow{g} & U \end{array}$$

```
def Co (C: precategory) (cod: C.C.ob) : U
 := Σ (dom: C.C.ob), C.C.hom dom cod

def Delta (C: precategory) (d: C.C.ob) : U₁
 := Σ (index: U), index -> Co C d

def Coverage (C: precategory): U₁
 := Σ (cod: C.C.ob) (fam: Delta C cod)
      (coverings: C.C.ob -> Delta C cod -> U),
      coverings cod fam

def site  (C: precategory): U₁
 := Σ (C: precategory), Coverage C
```

Definition 12. (Grothendieck Topology). Suppose category C has all pullbacks. Since C is small, a pretopology on C consists of families of sets of morphisms

$$\{\phi_\alpha : U_\alpha \to U\}, U \in C,$$

called covering families, such that following axioms hold: i) suppose that $\phi_\alpha : U_\alpha \to U$ is a covering family and that $\psi : V \to U$ is a morphism of C. Then the collection $V \times_U U_\alpha \to V$ is a cvering family for V. ii) If $\{\phi_\alpha : U_\alpha \to U\}$ is covering, and $\{\gamma_{\alpha,\beta} : W_{\alpha,\beta} \to U_\alpha\}$ is covering for all α, then the family of composites

$$W_{\alpha,\beta} \xrightarrow{\gamma_{\alpha,\beta}} U_\alpha \xrightarrow{\phi_\alpha} U$$

is covering; iii) The family $\{1 : U \to U\}$ is covering for all $U \in C$.

Definition 13. (Site). Site is a category having either a coverage, grothendieck topology, or sieves.

```
site (C: precategory): U
  = (C: precategory) * Coverage C
```

Definition 14. (Presheaf). Presheaf of a category C is a functor from opposite category to category of sets: $C^{op} \to \mathrm{Set}$.

```
presheaf (C: precategory): U
  = catfunctor (opCat C) Set
```

Definition 15. (Presheaf Category, **PSh**). Presheaf category **PSh** for a site C is category were objects are presheaves and morphisms are natural transformations of presheaf functors.

Definition 16. (Sheaf). Sheaf is a presheaf on a site. In other words a presheaf $\mathsf{F} : \mathsf{C}^{\mathrm{op}} \to \boldsymbol{Set}$ such that the cannonical map of inverse limit

$$\mathsf{F}(\mathsf{U}) \to \varprojlim_{\mathsf{V} \to \mathsf{U} \in \mathsf{R}} \mathsf{F}(\mathsf{V})$$

is an isomorphism for each covering sieve $\mathsf{R} \subset \mathsf{Hom}_{\mathsf{C}}(\_, \mathsf{U})$. Equivalently, all induced functions

$$\mathsf{Hom}_{\mathsf{C}}(\mathsf{Hom}_{\mathsf{C}}(\_, \mathsf{U}), \mathsf{F}) \to \mathsf{Hom}_{\mathsf{C}}(\mathsf{R}, \mathsf{F})$$

should be bejections.

```
sheaf (C: precategory): U
  = (S: site C)
  * presheaf S.1
```

Definition 17. (Sheaf Category, **Sh**). Sheaf category **Sh** is a category where objects are sheaves and morphisms are natural transformation of sheves. Sheaf category is a full subcategory of category of presheaves **PSh**.

Definition 18. (Grothendieck Topos). Topos is the category of sheaves $\mathbf{Sh}(\mathsf{C}, \mathsf{J})$ on a site C with topology J.

Theorem 2. (Giraud). A category C is a Grothiendieck topos iff it has following properties: i) has all finite limits; ii) has small disjoint coproducts stable under pullbacks; iii) any epimorphism is coequalizer; iv) any equivalence relation $\mathsf{R} \to \mathsf{E}$ is a kernel pair and has a quotient; v) any coequalizer $\mathsf{R} \to \mathsf{E} \to \mathsf{Q}$ is stably exact; vi) there is a set of objects that generates C.

Definition 19. (Geometric Morphism). Suppose that C and D are Grothendieck sites. A geometric morphism

$$\mathsf{f} : \mathbf{Sh}(\mathsf{C}) \to \mathbf{Sh}(\mathsf{D})$$

consist of functors $\mathsf{f}_* : \mathbf{Sh}(\mathsf{C}) \to \mathbf{Sh}(\mathsf{D})$ and $\mathsf{f}^* : \mathbf{Sh}(\mathsf{D}) \to \mathbf{Sh}(\mathsf{C})$ such that f^* is left adjoint to f_* and f^* preserves finite limits. The left adjoint f^* is called the inverse image functor, while f_* is called the direct image. The inverse image functor f^* is left and right exact in the sense that it preserves all finite colimits and limits, respectively.

Definition 20. (Cohesive Topos). A topos E is a cohesive topos over a base topos S, if there is a geometric morphism $(\mathsf{p}^*, \mathsf{p}_*) : \mathsf{E} \to \mathsf{S}$, such that: i) exists adjunction $\mathsf{p}^! \vdash \mathsf{p}_*$ and $\mathsf{p}^! \dashv \mathsf{p}_*$; ii) p^* and $\mathsf{p}^!$ are full faithful; iii) $\mathsf{p}_!$ preserves finite products.

This quadruple defines adjoint triple:

$$\int \dashv \flat \dashv \sharp$$

1.4 Elementary Topos

Giraud theorem was a synonymical topos definition involved only topos properties but not a site properties. That was step forward on predicative definition. The other step was made by Lawvere and Tierney, by removing explicit dependance on categorical model of set theory (as category of set is used in definition of presheaf). This information was hidden into subobject classifier which was well defined through categorical pullback and property of being cartesian closed (having lambda calculus as internal language).

Elementary topos doesn't involve 2-categorical modeling, so we can construct set topos without using functors and natural transformations (what we need in geometrical topos theory flavour). This flavour of topos theory more suited for logic needs rather that geometry, as its set properties are hidden under the predicative predicative pullback definition of subobject classifier rather that functorial notation of presheaf functor. So we can simplify proofs at the homotopy levels, not to lift everything to 2-categorical model.

Definition 21. (Monomorphism). An morphism $f : Y \to Z$ is a monic or mono if for any object X and every pair of parralel morphisms $g_1, g_2 : X \to Y$ the

$$f \circ g_1 = f \circ g_2 \to g_1 = g_2.$$

More abstractly, f is mono if for any X the $\mathrm{Hom}(X, \_)$ takes it to an injective function between hom sets $\mathrm{Hom}(X, Y) \to \mathrm{Hom}(X, Z)$.

```
mono (P: precategory) (Y Z: carrier P) (f: hom P Y Z): U
   = (X: carrier P) (g1 g2: hom P X Y)
  -> Path (hom P X Z) (compose P X Y Z g1 f)
                      (compose P X Y Z g2 f)
  -> Path (hom P X Y) g1 g2
```

Definition 22. (Subobject Classifier[**?**]). In category C with finite limits, a subobject classifier is a monomorphism $\mathsf{true} : 1 \to \Omega$ out of terminal object 1, such that for any mono $U \to X$ there is a unique morphism $\chi_U : X \to \Omega$ and pullback diagram:

$$\begin{array}{ccc} U & \xrightarrow{k} & 1 \\ \downarrow & & \downarrow{\scriptstyle \mathsf{true}} \\ X\Omega & \xrightarrow{\chi_U} & \Omega \end{array}$$

```
subobjectClassifier (C: precategory): U
  = (omega: carrier C)
  * (end: terminal C)
  * (trueHom: hom C end.1 omega)
  * (chi: (V X: carrier C) (j: hom C V X) -> hom C X omega)
  * (square: (V X: carrier C) (j: hom C V X) -> mono C V X j
    -> hasPullback C (omega,(end.1,trueHom),(X,chi V X j)))
  * ((V X: carrier C) (j: hom C V X) (k: hom C X omega)
    -> mono C V X j
    -> hasPullback C (omega,(end.1,trueHom),(X,k))
    -> Path (hom C X omega) (chi V X j) k)
```

Theorem 3. (Category of Sets has Subobject Classifier).

Definition 23. (Cartesian Closed Categories). The category C is called cartesian closed if exists all: i) terminals; ii) products; iii) exponentials. Note that this definition lacks beta and eta rules which could be found in embedding **MLTT**.

```
isCCC (C: precategory): U
  = (Exp:   (A B: carrier C) -> carrier C)
  * (Prod:  (A B: carrier C) -> carrier C)
  * (Apply: (A B: carrier C) -> hom C (Prod (Exp A B) A) B)
  * (P1:    (A B: carrier C) -> hom C (Prod A B) A)
  * (P2:    (A B: carrier C) -> hom C (Prod A B) B)
  * (Term:  terminal C)
  * unit
```

Theorem 4. (Category of Sets is cartesian closed). As you can see from exp and pro we internalize Π and Σ types as SET instances, the isSet predicates are provided with contractability. Exitense of terminals is proved by propPi. The same technique you can find in **MLTT** embedding.

```
cartesianClosure : isCCC Set
  = (expo,prod,appli,proj1,proj2,term,tt) where
    exp (A B: SET): SET = (A.1   -> B.1, setFun A.1 B.1 B.2)
    pro (A B: SET): SET = (prod A.1 B.1, setSig A.1 (\(_ : A.1)
                          -> B.1) A.2 (\(_ : A.1) -> B.2))
    expo:  (A B: SET) -> SET = \(A B: SET) -> exp A B
    prod:  (A B: SET) -> SET = \(A B: SET) -> pro A B
    appli: (A B: SET) -> hom Set (pro (exp A B) A) B
        = \(A B: SET) -> \(x:(pro(exp A B)A).1)-> x.1 x.2
    proj1: (A B: SET) -> hom Set (pro A B) A
        = \(A B: SET) (x: (pro A B).1) -> x.1
    proj2: (A B: SET) -> hom Set (pro A B) B
        = \(A B: SET) (x: (pro A B).1) -> x.2
    unitContr (x: SET) (f: x.1 -> unit) : isContr (x.1 -> unit)
      = (f, \(z: x.1 -> unit) -> propPi x.1 (\(_:x.1)->unit)
           (\(x:x.1) -> propUnit) f z)
    term: terminal Set = ((unit,setUnit),
           \(x: SET) -> unitContr x (\(z: x.1) -> tt))
```

Note that rules of cartesian closure forms a type theoretical langage called lambda calculus.

Definition 24. (Elementary Topos). Topos is a precategory which is cartesian closed and has subobject classifier.

```
Topos (cat: precategory) : U
  = (cartesianClosure: isCCC cat)
  * subobjectClassifier cat
```

Theorem 5. (Topos Definitions). Any Grothendieck topos is an elementary topos too. The proof is sligthly based on results of Giraud theorem.

Theorem 6. (Category of Sets forms a Topos). There is a cartesian closure and subobject classifier for a categoty of sets.

```
internal : Topos Set
         = (cartesianClosure, hasSubobject)
```

Theorem 7. (Freyd). Main theorem of topos theory[**?**]. For any topos C and any $\mathsf{b} : \mathrm{Ob}_{\mathsf{C}}$ relative category $\mathsf{C} \downarrow \mathsf{b}$ is also a topos. And for any arrow $\mathsf{f} : \mathsf{a} \to \mathsf{b}$ inverse image functor $\mathsf{f}^* : \mathsf{C} \downarrow \mathsf{b} \to \mathsf{c} \downarrow \mathsf{a}$ has left adjoint $\sum_{\mathsf{f}}$ and right adjoin $\prod_{\mathsf{f}}$.

Conclusion

We gave here constructive definition of topology as finite unions and intersections of open subsets. Then make this definition categorically compatible by introducing Grothendieck topology in three different forms: sieves, coverage, and covering families. Then we defined an elementary topos and introduce category of sets, and proved that **Set** is cartesian closed, has object classifier and thus a topos.

This intro could be considered as a formal introduction to topos theory (at least of the level of first chapter) and you may evolve this library to your needs or ask to help porting or developing your application of topos theory to a particular formal construction.