Issue XXXI: Abelian Categories

Namdak Tonpa

15 травня 2025 р.

Анотація

Ця стаття є оглядом абелевих категорій, введених Александром Гротендіком у 1957 році, як фундаментального інструменту гомологічної алгебри, алгебраїчної геометрії, теорії представлень, топологічної квантової теорії поля та теорії категорій. Ми розглядаємо формальне означення абелевих категорій, їхню роль у побудові похідних категорій і функторів, а також ключові застосування в різних галузях математики та фізики.

Зміст

1 Abelian Categories **2**
1.1 Означення абелевих категорій 2
1.2 Деталізоване формальне означення 2
1.3 Мотивація та застосування 4
1.4 Похідні категорії та функтори 4
1.5 Висновки 5

1 Abelian Categories

Абелеві категорії, вперше введені Александром Гротендіком у його статті 1957 року «Sur quelques points d'algèbre homologique» [1], стали основою для уніфікації гомологічної алгебри в різних математичних дисциплінах, таких як алгебраїчна геометрія, алгебраїчна топологія та теорія представлень. Вони забезпечують природне середовище для вивчення гомологій, когомологій, похідних категорій і функторів, що мають широке застосування в математиці та математичній фізиці.

1.1 Означення абелевих категорій

Абелеві категорії — це збагачене поняття категорії Сандерса-Маклейна поняттями нульового об'єкту, що одночасно ініціальний та термінальний, властивостями існування всіх добутків та кодобутків, ядер та коядер, а також, що всі мономорфізми і епіморфізми є ядрами і коядрами відповідно (тобто нормальними).

Формально, абелева категорія визначається наступним чином:

```
def isAbelian (C: precategory): U₁
 := Σ (zero:    hasZeroObject C)
      (prod:    hasAllProducts C)
      (coprod:  hasAllCoproducts C)
      (ker:     hasAllKernels C zero)
      (coker:   hasAllCokernels C zero)
      (monicsAreKernels:
         Π (A S: C.C.ob) (k: C.C.hom S A),
         Σ (B: C.C.ob) (f: C.C.hom A B),
         isKernel C zero A B S f k)
      (epicsAreCoKernels:
         Π (B S: C.C.ob) (k: C.C.hom B S),
         Σ (A: C.C.ob) (f: C.C.hom A B),
         isCokernel C zero A B S f k), U
```

Ця сигнатура включає: 1) існування нульового об'єкта; 2) існування всіх добутків; 3) існування всіх кодобутків; 4) існування всіх ядер; 5) існування всіх коядер; 6) властивість, що кожен мономорфізм є ядром; 7) властивість, що кожен епіморфізм є коядром.

1.2 Деталізоване формальне означення

Для чіткості наведемо ключові компоненти абелевої категорії в сучасному формалізмі, наприклад, у кубічній Агді, як описано в магістерській роботі

Девіда Еліндера 2021 року [2]:

```
module abelian where
import lib/mathematics/categories/category
import lib/mathematics/homotopy/truncation

def zeroObject(C: precategory) (X: C.C.ob): U₁
 := Σ (bot: isInitial C X) (top: isTerminal C X), U

def hasZeroObject (C: precategory) : U₁
 := Σ (ob: C.C.ob) (zero: zeroObject C ob), unit

def hasAllProducts (C: precategory) : U₁
 := Σ (product: C.C.ob -> C.C.ob -> C.C.ob)
      (π₁: Π (A B : C.C.ob), C.C.hom (product A B) A)
      (π₂: Π (A B : C.C.ob), C.C.hom (product A B) B), U

def hasAllCoproducts (C: precategory) : U₁
 := Σ (coproduct: C.C.ob -> C.C.ob -> C.C.ob)
      (σ₁: Π (A B : C.C.ob), C.C.hom A (coproduct A B))
      (σ₂: Π (A B : C.C.ob), C.C.hom B (coproduct A B)), U

def isMonic (P: precategory) (Y Z : P.C.ob) (f : P.C.hom Y Z) : U
 := Π (X : P.C.ob) (g1 g2 : P.C.hom X Y),
    Path (P.C.hom X Z) (P.P.∘ X Y Z g1 f) (P.P.∘ X Y Z g2 f)
 -> Path (P.C.hom X Y) g1 g2

def isEpic (P : precategory) (X Y : P.C.ob) (f : P.C.hom X Y) : U
 := Π (Z : P.C.ob) (g1 g2 : P.C.hom Y Z),
    Path (P.C.hom X Z) (P.P.∘ X Y Z f g1) (P.P.∘ X Y Z f g2)
 -> Path (P.C.hom Y Z) g1 g2

def kernel (C: precategory) (zero: hasZeroObject C)
    (A B S: C.C.ob) (f: C.C.hom A B) : U₁
 := Σ (k: C.C.hom S A) (monic: isMonic C S A k), unit

def cokernel (C: precategory) (zero: hasZeroObject C)
    (A B S: C.C.ob) (f: C.C.hom A B)  : U₁
 := Σ (k: C.C.hom B S) (epic: isEpic C B S k), unit

def isKernel (C: precategory) (zero: hasZeroObject C)
    (A B S: C.C.ob) (f: C.C.hom A B) (k: C.C.hom S A) : U₁
 := Σ (ker: kernel C zero A B S f), Path (C.C.hom S A) ker.k k

def isCokernel (C: precategory) (zero: hasZeroObject C)
    (A B S: C.C.ob) (f: C.C.hom A B) (k: C.C.hom B S) : U₁
 := Σ
(coker: cokernel C zero A B S f), Path (C.C.hom B S) coker.k k

def hasKernel (C: precategory) (zero: hasZeroObject C)
    (A B: C.C.ob) (f: C.C.hom A B) : U₁
 := ‖_‖₋₁ (Σ (monic: isMonic C A B f), unit)

def hasCokernel (C: precategory) (zero: hasZeroObject C)
    (A B: C.C.ob) (f: C.C.hom A B) : U₁
 := ‖_‖₋₁ (Σ (epic: isEpic C A B f), unit)

def hasAllKernels (C : precategory) (zero: hasZeroObject C) : U₁
```

```
 := Σ (A B : C.C.ob) (f : C.C.hom A B), hasKernel C zero A B f

def hasAllCokernels (C : precategory) (zero: hasZeroObject C) : U₁
 := Σ (A B : C.C.ob) (f : C.C.hom A B), hasCokernel C zero A B f
```

Ці означення уточнюють поняття нульового об'єкта, добутків, кодобутків, мономорфізмів, епіморфізмів, ядер і коядер, необхідних для абелевих категорій.

1.3 Мотивація та застосування

Абелеві категорії мають численні застосування в різних галузях математики та фізики. Ось п'ять ключових напрямів:

1) Гомологічна алгебра: абелеві категорії забезпечують основу для гомологічної алгебри, яка вивчає властивості груп гомології та когомології. Теорія похідних функторів, фундаментальний інструмент гомологічної алгебри, базується на понятті абелевої категорії.

2) Алгебраїчна геометрія: абелеві категорії використовуються для вивчення когомологій пучка, що є потужним інструментом для розуміння геометричних властивостей алгебраїчних многовидів. Зокрема, категорія пучків абелевих груп на топологічному просторі є абелевою категорією.

3) Теорія представлень: абелеві категорії виникають у теорії представлень, яка досліджує алгебраїчні структури, пов'язані з симетріями. Наприклад, категорія модулів над кільцем є абелевою категорією.

4) Топологічна квантова теорія поля: абелеві категорії відіграють центральну роль у топологічній квантовій теорії поля, де вони виникають як категорії граничних умов для певних типів теорій топологічного поля.

5) Теорія категорій: абелеві категорії є важливим об'єктом дослідження в теорії категорій, зокрема для вивчення адитивних функторів. Рекомендується робота Бакура і Деляну «Вступ в теорію категорій та функторів» [3] для поглибленого ознайомлення.

1.4 Похідні категорії та функтори

Абелеві категорії забезпечують природну основу для гомологічної алгебри, яка є розділом алгебри, що має справу з алгебраїчними властивостями груп гомологій та когомологій. Зокрема, абелеві категорії створюють сеттінг, де можна визначити поняття похідних категорій і похідних функторів.

Основна ідея похідних категорій полягає в тому, щоб ввести нову категорію, яка побудована з абелевої категорії шляхом «інвертування» певних морфізмів, майже так само, як будується поле часток на області цілісності. Похідна категорія абелевої категорії фіксує «правильне» поняття гомологічних і когомологічних груп і забезпечує потужний інструмент для вивчення алгебраїчних властивостей цих груп.

Похідні функтори є фундаментальним інструментом гомологічної алгебри, і їх можна визначити за допомогою концепції похідної категорії. Основна ідея похідних функторів полягає в тому, щоб взяти функтор, який визначено в абелевій категорії, і «підняти» його до функтора, який визначений у похідній категорії. Похідний функтор потім використовується для обчислення вищих груп гомології та когомології об'єктів в абелевій категорії.

Використання похідних категорій і функторів зробило революцію у вивченні гомологічної алгебри, і це призвело до багатьох важливих застосувань в алгебраїчній геометрії, топології та математичній фізиці. Наприклад, похідні категорії використовувалися для доведення фундаментальних результатів алгебраїчної геометрії, таких як знаменита теорема Гротендіка-Рімана-Роха. Вони також використовувалися для вивчення дзеркальної симетрії в теорії суперструн.

1.5 Висновки

Абелеві категорії, введені Гротендіком, є фундаментальним інструментом сучасної математики, що забезпечує уніфікований підхід до гомологічної алгебри, алгебраїчної геометрії, теорії представлень, топологічної квантової теорії поля та теорії категорій. Їхня роль у побудові похідних категорій і функторів відкрила нові можливості для вивчення гомологій і когомологій, а також їхніх застосувань у математиці та фізиці. Подальший розвиток теорії абелевих категорій, зокрема в контексті унівалентної теорії типів, як показано в роботі Еліндера [2], обіцяє нові перспективи для формальної математики та комп'ютерних наук.

Література

[1] А. Гротендік, *Sur quelques points d'algèbre homologique*, 1957.

[2] Д. Еліндер, *Дослідження абелевих категорій і унівалентної теорії типів*, магістерська робота, 2021.

[3] І. Бакур, А. Деляну, *Вступ в теорію категорій та функторів.*