Випуск IV: Вищі індуктивні типи

Максим Сохацький [1]

[1] Національний технічний університет України
Київський політехнічний інститут імені Ігоря Сікорського

4 травня 2019

Анотація

CW-комплекси є ключовими як в теорії гомотопій так і в теорії гомотопічних типів (HoTT) і кодуються в кубічних системах доведення теоремяк вищі індуктивні типи (HIT). Подібно до рекурсивних дерев для (ко)індуктивних типів, HIT представляють CW-комплекси. Ми досліджуємо мотивацію HIT, їхню топологічну роль та реалізацію в Agda Cubical, зосереджуючись на базових примітивах гомотопічної теорії.

Ключові слова: Клітинна топологія, Кубічна теорія типів, Вищі індуктивні типи

Зміст

1 CW-комплекси

CW-комплекси — це простори, побудовані шляхом приєднання клітин зростаючої розмірності. У HoTT вони кодуються як HITs, де клітини є конструкторами для точок і шляхів.

Означення 1. (Приєднання клітини). Приєднання n-клітини до простору X вздовж $f : S^{n-1} \to X$ є розшарованою сумою:

$$\begin{tikzcd} S^{n-1} \arrow[r, "f"] \arrow[d, "\iota"] & X \arrow[d, "j"] \\ D^n \arrow[r, "g"] & X \cup_f D^n \end{tikzcd}$$

Тут $\iota : S^{n-1} \hookrightarrow D^n$ — включення межі, а $X \cup_f D^n$ — розшарована сума, що приклеює n-клітину до X через f. Результат залежить від гомотопічного класу f.

Означення 2. (CW-комплекс). CW-комплекс — це простір X, побудований індуктивно шляхом приєднання клітин, із скелетною фільтрацією:

- (-1)-скелет: $X_{-1} = \varnothing$.
- Для $n \geq 0$, n-скелет X_n отримується приєднанням n-клітин до X_{n-1}. Для індексів J_n та відображень $\{f_j : S^{n-1} \to X_{n-1}\}_{j \in J_n}$, X_n є розшарованою сумою:

  $$\begin{tikzcd} \coprod_{j \in J_n} S^{n-1} \arrow[r, "\coprod f_j"] \arrow[d, "\coprod \iota_j"] & X_{n-1} \arrow[d, "i_n"] \\ \coprod_{j \in J_n} D^n \arrow[r, "\coprod g_j"] & X_n \end{tikzcd}$$

  де $\coprod_{j \in J_n} S^{n-1}$, $\coprod_{j \in J_n} D^n$ — диз'юнктні об'єднання, а $i_n : X_{n-1} \hookrightarrow X_n$ — включення.
- X — коліміта:

  $$\varnothing = X_{-1} \hookrightarrow X_0 \hookrightarrow X_1 \hookrightarrow \ldots \hookrightarrow X,$$

  де X_n — n-скелет, а $X = \mathrm{colim}_{n \to \infty} X_n$. Послідовність є скелетною фільтрацією.

У HoTT CW-комплекси є HITs із конструкторами для клітин і шляхів для приклеювання.

Приклад 1. (Сфера як CW-комплекс). n-сфера S^n — це CW-комплекс з однією 0-клітиною та однією n-клітиною:

- $X_0 = \{\text{base}\}$, точка.
- $X_k = X_0$ для $0 < k < n$, без додаткових клітин.
- X_n: Приєднання n-клітини до $X_{n-1} = \{\text{base}\}$ вздовж $f : S^{n-1} \to \{\text{base}\}$: Конструктор cell приклеює межі до base, отримуючи S^n.

1.1 Мотивація вищих індуктивних типів

HITs у HoTT дозволяють безпосередньо кодувати топологічні простори, такі як CW-комплекси. У теорії гомотопій простори будуються шляхом приклеювання клітин через відображення приєднання. HoTT розглядає типи як простори, елементи як точки, а рівності як шляхи, що робить HITs природним вибором. Стандартні індуктивні типи не можуть захопити вищі гомотопії, але HITs дозволяють конструктори для точок і шляхів. Наприклад, коло S^1 (Означення 2) має базову точку і петлю, кодуючи його фундаментальну групу $\mathbb{Z}$. HITs уникають використання множинних фактор-просторів, зберігаючи синтетичну природу HoTT. У кубічній теорії типів шляхи є інтервалами (наприклад, $< i >$) з обчислювальним змістом, на відміну від пропозиційних рівностей, що забезпечує ефективну перевірку типів у таких інструментах, як Agda Cubical.

1.2 HITs з конструкторами нескінченності

Деякі HITs потребують нескінченних конструкторів для просторів, таких як простори Ейленберга-МакЛейна або нескінченна сфера S^∞.

```
def S∞ : U
 := inductive { base
              | loop (n: Nat) <i> [ (i=0) -> base, (i=1) -> base ]
              }
```

Виклики включають перевірку типів, обчислення та виразність.

Agda Cubical використовує кубічні примітиви для роботи з HITs, підтримуючи нескінченні конструктори через індексовані HITs.

```
def HIT-∞ (A: U) : U
 := inductive { point : HIT-∞ A
              | path : (n: Nat) -> PathP (λ i, HIT-∞ A) point point
              }
```

2 Вищі індуктивні типи

CW-комплекси є центральними в HoTT і з’являються в кубічних перевірялниках типів як HITs. На відміну від індуктивних типів (рекурсивних дерев), HITs кодують CW-комплекси, захоплюючи точки (0-клітини) та вищі шляхи (n-клітини). Означення HIT визначає CW-комплекс через кубічну композицію, початкову алгебру в кубічній моделі.

2.1 Суспензія

Суспензія ΣA типу A — це вищий індуктивний тип, який конструює новий тип, додаючи дві точки, звані полюсами, і шляхи, що з'єднують кожну точку A з цими полюсами. Це фундаментальна конструкція в теорії гомотопій, яка часто використовується для зсуву гомотопічних груп, наприклад, для отримання S^{n+1} з S^n.

Означення 3. (Формування) Для типу $A : \mathcal{U}$, суспензія $\Sigma A : \mathcal{U}$.

Означення 4. (Введення) Суспензія генерується такою вищою індуктивною композиційною структурою: $\text{north}, \text{south} : \Sigma A$ та $\text{merid} : (a : A) \to \text{north} \equiv \text{south}$.

```
def Σ (A: U) : U
 := inductive { north
              | south
              | merid (a: A) : Path (ΣA) north south
              }
```

Теорема 1. (Елімінація) Для сімейства типів $B : \Sigma A \to \mathcal{U}$, точок $n : B(\text{north})$, $s : B(\text{south})$, і сімейства залежних шляхів

$$m : (a : A) \to \text{PathOver}(B, \text{merid}(a), n, s),$$

існує залежне відображення $\text{Ind}_{\Sigma A} : (x : \Sigma A) \to B(x)$, таке що:

$$\begin{cases} \text{Ind}_{\Sigma A}(\text{north}) = n \\ \text{Ind}_{\Sigma A}(\text{south}) = s \\ \text{Ind}_{\Sigma A}(\text{merid}(a, i)) = m(a, i) \end{cases}$$

```
def PathOver (B: Σ A -> U) (a: A) (n: B north) (s: B south) : U
  := PathP (λ i , B (merid a @ i)) n s

def IndΣA (A: U) (B: Σ A -> U) (n: B north) (s: B south)
     (m: (a: A) -> PathOver B (merid a) n s) : (x: Σ A) -> B x
 := split { north -> n | south -> s | merid a @ i -> m a @ i }
```

Теорема 2. (Обчислення)

```
def Σ-β (A: U) (B: ΣA -> U) (n: B north) (s: B south)
     (m: (a: A) -> PathOver B (merid a) n s) (x: Σ A)
  : Path (B x) (Σ-I A B n s m x)
     split { north -> n | south -> s | merid a @ i -> m a @ i } x
  = idp (B x) (IndΣA A B n s m x)
```

Теорема 3. (Унікальність) Будь-які два відображення $h_1, h_2 : (x : \Sigma A) \to B(x)$ є гомотопними, якщо вони збігаються на north, south і merid, тобто, якщо $h_1(\text{north}) = h_2(\text{north})$, $h_1(\text{south}) = h_2(\text{south})$, і $h_1(\text{merid } a) = h_2(\text{merid } a)$ для всіх $a : A$.

2.2 Розшарована сума

Розшарована сума (амальгама) — це вищий індуктивний тип, що конструює тип шляхом склеювання двох типів A і B вздовж спільного типу C через відображення $f : C \to A$ і $g : C \to B$. Це фундаментальна конструкція в теорії гомотопій, використовується для моделювання приєднання клітин і кофібрантних об'єктів, узагальнюючи топологічне поняття розшарованої суми.

Означення 5. (Формування) Для типів $A, B, C : \mathcal{U}$ і відображень $f : C \to A$, $g : C \to B$, існує розшарована суса $\sqcup(A, B, C, f, g) : \mathcal{U}$.

Означення 6. (Введення) Розшарована сума генерується такою вищою індуктивною композиційною структурою:

$$\begin{cases} \mathrm{po}_1 : A \to \sqcup(A, B, C, f, g) \\ \mathrm{po}_2 : B \to \sqcup(A, B, C, f, g) \\ \mathrm{po}_3 : (c : C) \to \mathrm{po}_1(f(c)) \equiv \mathrm{po}_2(g(c)) \end{cases}$$

```
def ⊔ (A B C: U) (f: C -> A) (g: C -> B) : U
 := inductive { po₁ (_: A)
              | po₂ (_: B)
              | po₃ (c: C) : po₁(f(c)) ≡ po₂(g(c))
              }
```

Теорема 4. (Елімінація) Для типу $D : \mathcal{U}$, відображень $u : A \to D$, $v : B \to D$, і сімейства шляхів $p : (c : C) \to u(f(c)) \equiv v(g(c))$, існує відображення $\mathrm{Ind}_{\sqcup} : \sqcup(A, B, C, f, g) \to D$, таке що:

$$\begin{cases} \mathrm{Ind}_{\sqcup}(\mathrm{po}_1(a)) = u(a) \\ \mathrm{Ind}_{\sqcup}(\mathrm{po}_2(b)) = v(b) \\ \mathrm{Ind}_{\sqcup}(\mathrm{po}_3(c, i)) = p(c, i) \end{cases}$$

```
def PathOver (A B C : U) (f : C → A) (g : C → B)
    (D : ⊔ A B C f g → U)
    (c : C) (u : D (po₁ (f c))) (v : D (po₂ (g c))) : U
 := PathP (λ i, D (po₃ c i)) u v

def Ind⊔ : (A B C : U) (f : C → A) (g : C → B)
    (D : ⊔ A B C f g → U)
    (u : (a : A) → D (po₁ a))
    (v : (b : B) → D (po₂ b))
    (p : (c : C) → PathOver D c (u (f c)) (v (g c)))
  : (x : ⊔ A B C f g) → D x
 := split { po₁ a → u a | po₂ b → v b | po₃ c @ i → p c @ i }
```

Теорема 5. (Обчислення) Для $x : \sqcup(A, B, C, f, g)$,

$$\begin{cases} \mathrm{Ind}_{\sqcup}(\mathrm{po}_1(a)) \equiv u(a) \\ \mathrm{Ind}_{\sqcup}(\mathrm{po}_2(b)) \equiv v(b) \\ \mathrm{Ind}_{\sqcup}(\mathrm{po}_3(c, i)) \equiv p(c, i) \end{cases}$$

Теорема 6. (Унікальність) Будь-які два відображення $u, v : \sqcup(A, B, C, f, g) \to D$ є гомотопними, якщо вони збігаються на po_1, po_2 і po_3, тобто, якщо $u(\mathrm{po}_1(a)) = v(\mathrm{po}_1(a))$ для всіх $a : A$, $u(\mathrm{po}_2(b)) = v(\mathrm{po}_2(b))$ для всіх $b : B$, і $u(\mathrm{po}_3(c)) = v(\mathrm{po}_3(c))$ для всіх $c : C$.

Приклад 2. (Приєднання клітини) Розшарована сума моделює приєднання n-клітини до простору X. Дано $f : S^{n-1} \to X$ і включення $g : S^{n-1} \to D^n$, розшарована сума $\sqcup(X, D^n, S^{n-1}, f, g)$ є простором $X \cup_f D^n$, що приклеює n-диск до X вздовж f.

$$\begin{tikzcd} S^{n-1} \arrow[r, "f"] \arrow[d, "g"] & X \arrow[d] \\ D^n \arrow[r] & X \cup_f D^n \end{tikzcd}$$

2.3 Сфери

Сфери — це вищі індуктивні типи (HITs) із шляхами вищої розмірності, що представляють фундаментальні топологічні простори.

Означення 7. (Точкові n-сфери) n-сфера S^n визначається рекурсивно як тип у всесвіті $\mathcal{U}$ за допомогою загальної рекурсії за розмірностями:

$$\mathbb{S}^n := \begin{cases} \mathrm{point} : \mathbb{S}^n, \\ \mathrm{surface} : < i_1, ...i_n > [\ (i_1 = 0) \to point, (i_1 = 1) \to point,\ ... \\ \qquad\qquad\qquad\qquad (i_n = 0) \to point, (i_n = 1) \to point\ ] \end{cases}$$

Означення 8. (Суспендовані n-сфери) n-сфера S^n визначається рекурсивно як тип у всесвіті $\mathcal{U}$ за допомогою загальної рекурсії над натуральними числами $\mathbb{N}$. Для кожного $n \in \mathbb{N}$, тип $S^n : \mathcal{U}$ визначається так:

$$\mathbb{S}^n := \begin{cases} S^0 = \mathbf{2}, \\ S^{n+1} = \Sigma(S^n). \end{cases}$$

```
def sphere : ℕ → U := ℕ-iter U 2 Σ
```

Ця ітеративна означення застосовує функтор суспензії Σ до базового типу $\mathbf{2}$ (0-сфера) n разів, щоб отримати S^n.

Приклад 3. (Сфера як CW-комплекс) n-сфера S^n може бути побудована як CW-комплекс з однією 0-клітиною та однією n-клітиною:

$$\begin{cases} X_0 = \{\mathrm{base}\},\ \text{одна точка} \\ X_k = X_0 \text{ для } 0 < k < n,\ \text{без додаткових клітин} \\ X_n : \text{Приєднання } n\text{-клітини до } X_{n-1} = \{\mathrm{base}\} \text{ вздовж } f : S^{n-1} \to \{\mathrm{base}\} \end{cases}$$

Конструктор cell приклеює межу n-клітини до базової точки, отримуючи тип S^n.

2.4 Хаб і спиці

Конструкція хаб і спиці $\odot$ визначає n-відсікання, гарантуючи, що тип не має нетривіальних гомотопічних груп вище розмірності n. Вона моделює тип як CW-комплекс із хабом (центральною точкою) і спицями (шляхами до точок).

Означення 9. (Хаб і спиці) Для типів $S, A : \mathcal{U}$, тип хаб і спиці $\odot(S, A) : \mathcal{U}$.

$$
\begin{cases}
\text{base} : A \to \odot(S, A) \\
\text{hub} : (S \to \odot(S, A)) \to \odot(S, A) \\
\text{spoke} : (f : S \to \odot(S, A)) \to (s : S) \to \text{hub}(f) \equiv f(s) \\
\text{hubEq} : (x, y : A) \to (p : S \to x \equiv y) \to \text{base}(x) \equiv \text{base}(y) \\
\text{spokeEq} : (x, y : A) \to (p : S \to x \equiv y) \to (s : S) \to \text{hubEq}(x, y, p) \equiv \text{base}(p(s))
\end{cases}
$$

```
data hubSpokes (S A: U)
   = base (x: A)
   | hub (f: S -> hubSpokes S A)
   | spoke (f: S -> hubSpokes S A) (s:S)
     <i> [ (i=0) -> hub f , (i=1) -> f s ]
   | hubEq (x y: A) (p: S -> Path A x y)
     <i> [ (i=0) -> base x , (i=1) -> base y ]
   | spokeEq (x y: A) (p: S -> Path A x y) (s: S)
     <i> [ (i=0) -> hubEq x y p , (i=1) -> base (p s) ]
```

Теорема 7. (Елімінація hubSpokes) Для типу $B : \mathcal{U}$, відображення $g : A \to B$, точки $h : (S \to \text{hubSpokes}\ SA) \to B$, і відображень шляхів, що забезпечують когерентність, існує $\text{rec}_{\text{hubSpokes}} : \text{hubSpokes}\ SA \to B$, таке що $\text{rec}_{\text{hubSpokes}}(\text{base}\ x) = g(x)$ і $\text{rec}_{\text{hubSpokes}}(\text{hub}\ f) = h(f)$.

2.5 Відсікання множин

Відсікання множин (0-відсікання), позначене $\|A\|_0$, гарантує, що тип є множиною, з гомотопічними групами, що зникають вище розмірності 0.

Означення 10. (Відсікання множин) Для $A : \mathcal{U}$, $\|A\|_0 : \mathcal{U}$.

$$
\begin{cases}
\text{inc} : A \to \|A\|_0 \\
\text{squash} : (a, b : \|A\|_0) \to (p, q : a \equiv b) \to p \equiv q
\end{cases}
$$

```
data setTrunc (A: U)
   = inc (a: A)
   | squash (a b: setTrunc A) (p q: Path (setTrunc A) a b)
     <i j> [ (i = 0) -> p @ j, (i = 1) -> q @ j,
             (j = 0) -> a,     (j = 1) -> b ]
```

Теорема 8. (Елімінація $\|A\|_0$) Для множини $B : \mathcal{U}$ (тобто $\text{isSet}(B)$), відображення $f : A \to B$, існує $\text{setTruncRec} : \|A\|_0 \to B$, таке що $\text{setTruncRec}(\text{inc}(a)) = f(a)$.

2.6 Відсікання групоїдів

Відсікання групоїдів (1-відсікання), позначене $\|A\|_1$, гарантує, що тип є 1-групоїдом, з гомотопічними групами, що зникають вище розмірності 1.

Означення 11. (Відсікання групоїдів) Для $A : \mathcal{U}$, $\|A\|_1 : \mathcal{U}$.

$$\begin{cases} \text{inc} : A \to \|A\|_1 \\ \text{squash} : (a, b : \|A\|_1) \to (p, q : a \equiv b) \to (r, s : p \equiv q) \to r \equiv s \end{cases}$$

```
data grpdTrunc (A: U)
   = inc (a: A)
   | squash (a b: grpdTrunc A)
            (p q: Path (grpdTrunc A) a b)
            (r s: Path (Path (grpdTrunc A) a b) p q)
     <i j k> [ (i = 0) -> r @ j @ k, (i = 1) -> s @ j @ k,
               (j = 0) -> p @ k,     (j = 1) -> q @ k,
               (k = 0) -> a,         (k = 1) -> b ]
```

Теорема 9. (Елімінація $\|A\|_1$) Для 1-групоїда $B : \mathcal{U}$ (тобто isGroupoid(B)), відображення $f : A \to B$, існує grpdTruncRec : $\|A\|_1 \to B$, таке що grpdTruncRec(inc(a)) = $f(a)$.

2.7 Фактор-простори множин

Фактор-простори множин конструюють тип A, факторизований за відношенням $R : A \to A \to \mathcal{U}$, гарантуючи, що результат є множиною.

Означення 12. (Квоцієнт множин) Для типу $A : \mathcal{U}$ і відношення $R : A \to A \to \mathcal{U}$, фактор-простір множин setQuot $AR : \mathcal{U}$ визначається конструкторами:

$$\begin{cases} \text{quotient} : A \to \text{setQuot}(A, R) \\ \text{identification} : (a, b : A) \to Rab \to \text{quotient}(a) \equiv \text{quotient}(b) \\ \text{trunc} : (a, b : \text{setQuot}(A, R)) \to (p, q : a \equiv b \to p \equiv q \end{cases}$$

```
data setQuot (A: U) (R: A -> A -> U)
   = quotient (a: A)
   | identification (a b: A) (r: R a b)
     <i> [ (i=0) -> quotient a, (i=1) -> quotient b ]
   | trunc (a b : setQuot A R) (p q : Path (setQuot A R) a b)
     <i j> [ (i = 0) -> p @ j , (i = 1) -> q @ j ,
             (j = 0) -> a ,     (j = 1) -> b ]
```

Теорема 10. (Елімінація setQuot) Для сімейства типів B : setQuot $AR \to \mathcal{U}$ з isSet(Bx), і відображень $f : (x : A) \to B(\text{quotient } x)$, $g : (a, b : A) \to (r : Rab) \to \text{PathP}(< i > B(\text{idq } abr@i))(fa)(fb)$, існує setQuotElim : $\Pi_{x:\text{setQuot } AR} B(x)$, таке що setQuotElim(quotient a) = fa.

2.8 Фактор-простори групоїдів

Фактор-простори групоїдів розширюють фактор-простори множин для створення 1-групоїда, включаючи конструктори вищих шляхів.

Означення 13. (Фактор-простір групоїдів) Для типу $A : \mathcal{U}$ і відношення $R : A \to A \to \mathcal{U}$, квоцієнт групоїдів grpdQuot $AR : \mathcal{U}$ включає конструктори для точок, шляхів і вищих шляхів, що забезпечують структуру 1-групоїда. (Примітка: Повне означення потребує додаткової структури, частково опущено для стислості.)

2.9 Букет

Букет двох точкових типів A і B, позначена $A \vee B$, є вищим індуктивним типом (HIT), який представляє об'єднання A і B з ідентифікованими базовими точками. Топологічно вона відповідає $A \times \{y_0\} \cup \{x_0\} \times B$, де x_0 і y_0 — базові точки A і B, відповідно.

Означення 14. (Формування) Для точкових типів A, B : pointed, Букет Wedge $AB : \mathcal{U}$.

Означення 15. (Введення) Букет генерується такою вищою індуктивною композиційною структурою:

$$
\begin{cases}
\text{winl} : A.1 \to \text{Wedge } AB \\
\text{winr} : B.1 \to \text{Wedge } AB \\
\text{wglue} : \text{Path}_{\text{Wedge } AB}(\text{winl } A.2, \text{winr } B.2)
\end{cases}
$$

```
data Wedge (A : pointed) (B : pointed)
   = winl (a : A.1)
   | winr (b : B.1)
   | wglue <x> [ (x = 0) -> winl A.2 , (x = 1) -> winr B.2 ]
```

Теорема 11. (Елімінація) Для типу $C : \mathcal{U}$, відображень $f : A.1 \to C$, $g : B.1 \to C$, і шляху $p : \text{Path}_C(f(A.2), g(B.2))$, існує відображення WedgeRec : Wedge $AB \to C$, таке що:

$$
\begin{cases}
\text{WedgeRec}(\text{winl } a) = f(a) \\
\text{WedgeRec}(\text{winr } b) = g(b) \\
\text{WedgeRec}(\text{wglue } @\, x) = p\, @\, x
\end{cases}
$$

```
WedgeRec (A B : pointed) (C : U) (f : A.1 -> C) (g : B.1 -> C)
         (p : Path C (f A.2) (g B.2))
   : Wedge A B -> C
   = split
     winl a -> f a
     winr b -> g b
     wglue @ x -> p @ x
```

Теорема 12. (Обчислення) Для $z : \text{Wedge } AB$,

$$\begin{cases} \text{WedgeRec}(\text{winl } a) \equiv f(a) \\ \text{WedgeRec}(\text{winr } b) \equiv g(b) \\ \text{WedgeRec}(\text{wglue } @\, x) \equiv p \,@\, x \end{cases}$$

Теорема 13. (Унікальність) Будь-які два відображення $h_1, h_2 : \text{Wedge } AB \to C$ є гомотопними, якщо вони збігаються на winl, winr і wglue, тобто, якщо $h_1(\text{winl } a) = h_2(\text{winl } a)$ для всіх $a : A.1$, $h_1(\text{winr } b) = h_2(\text{winr } b)$ для всіх $b : B.1$, і $h_1(\text{wglue}) = h_2(\text{wglue})$.

2.10 Смеш-добуток

Смеш-добуток двох точкових типів A і B, позначений $A \wedge B$, є вищим індуктивним типом, який квоцієнтує продукт $A \times B$ за клинною сумою $A \vee B$. Він представляє простір $A \times B / (A \times \{y_0\} \cup \{x_0\} \times B)$, зводячи букет до однієї точки.

Означення 16. (Формування) Для точкових типів A, B : pointed, Смеш-добуток $\text{Smash } AB : \mathcal{U}$.

Означення 17. (Введення) Смеш-добуток генерується такою вищою індуктивною композиційною структурою:

$$\begin{cases} \text{spair} : A.1 \to B.1 \to \text{Smash } AB \\ \text{smash} : (a : A.1) \to (b : B.1) \to \text{Path}_{\text{Smash } AB}(\text{spair } a\, B.2, \text{spair } A.2\, b) \\ \text{smashpt} : \text{Path}_{\text{Smash } AB}(\text{smash } A.2\, B.2, \text{spair } A.2\, B.2) \end{cases}$$

```
data Smash (A : pointed) (B : pointed)
   = spair (a : A.1) (b : B.1)
   | smash (a : A.1) (b : B.1) <x> [(x=0) -> spair a B.2, (x=1) -> spair A.2 b]
   | smashpt <x y> [(x=0) -> smash A.2 B.2 @ y,
                    (x=1) -> spair A.2 B.2,
                    (y=0) -> spair A.2 B.2,
                    (y=1) -> spair A.2 B.2]
```

Теорема 14. (Елімінація) Для типу $C : \mathcal{U}$, відображення $f : A.1 \to B.1 \to C$, шляхів $g : (a : A.1) \to (b : B.1) \to \text{Path}_C(f a\, B.2, f A.2\, b)$, і 2-шляху $h : \text{Path}_{\text{Path}_{\text{Smash } AB}(f A.2\, B.2, f A.2\, B.2)}(g A.2\, B.2, \text{idp } (f A.2\, B.2))$, існує відображення $\text{SmashRec} : \text{Smash } AB \to C$, таке що:

$$\begin{cases} \text{SmashRec}(\text{spair } a\, b) = f(a, b) \\ \text{SmashRec}(\text{smash } a\, b \,@\, x) = g(a, b) \,@\, x \\ \text{SmashRec}(\text{smashpt } @\, x \,@\, y) = h \,@\, x \,@\, y \end{cases}$$

Теорема 15. (Обчислення) Для $z : \text{Smash } AB$,

$$\begin{cases} \text{SmashRec}(\text{spair } a\, b) \equiv f(a, b) \\ \text{SmashRec}(\text{smash } a\, b \,@\, x) \equiv g(a, b) \,@\, x \\ \text{SmashRec}(\text{smashpt } @\, x \,@\, y) \equiv h \,@\, x \,@\, y \end{cases}$$

Теорема 16. (Унікальність) Будь-які два відображення $h_1, h_2 : \mathrm{Smash}\ AB \to C$ є гомотопними, якщо вони збігаються на spair, smash і smashpt.

Приклад 4. (Смеш-добуток сфер) Смеш-добуток $S^1 \wedge S^1$ є гомотопічно еквівалентним S^2, оскільки він квоцієнтує тор $S^1 \times S^1$ за клин $S^1 \vee S^1$, зводячи базові точки та їхні волокна.

2.11 З'єднання

З'єднання двох типів A і B, позначене $A * B$, є вищим індуктивним типом, який конструює тип шляхом з'єднання кожної точки A з кожною точкою B через шлях. Топологічно воно відповідає з'єднанню просторів, формуючи простір, що інтерполює між A і B.

Означення 18. (Формування) Для типів $A, B : \mathcal{U}$, з'єднання $\mathrm{Join}\ AB : \mathcal{U}$.

Означення 19. (Введення) З'єднання генерується такою вищою індуктивною композиційною структурою:

$$\begin{cases} \mathrm{joinl} : A \to \mathrm{Join}\ AB \\ \mathrm{joinr} : B \to \mathrm{Join}\ AB \\ \mathrm{join} : (a : A) \to (b : B) \to \mathrm{Path}_{\mathrm{Join}\ AB}(\mathrm{joinl}\ a, \mathrm{joinr}\ b) \end{cases}$$

```
data Join (A : U) (B : U)
   = joinl (a : A)
   | joinr (b : B)
   | join (a:A) (b:B) <i> [(i=0) -> joinl a, (i=1) -> joinr b]
```

Теорема 17. (Елімінація) Для типу $C : \mathcal{U}$, відображень $f : A \to C$, $g : B \to C$, і сімейства шляхів $h : (a : A) \to (b : B) \to \mathrm{Path}_C(fa, gb)$, існує відображення $\mathrm{JoinRec} : \mathrm{Join}\ AB \to C$, таке що:

$$\begin{cases} \mathrm{JoinRec}(\mathrm{joinl}\ a) = f(a) \\ \mathrm{JoinRec}(\mathrm{joinr}\ b) = g(b) \\ \mathrm{JoinRec}(\mathrm{join}\ a\, b\, @\, i) = h(a, b)\, @\, i \end{cases}$$

```
JoinRec (A B C : U) (f : A -> C) (g : B -> C)
        (h : (a : A) -> (b : B) -> Path C (f a) (g b))
   : Join A B -> C
   = split
     joinl a -> f a
     joinr b -> g b
     join a b @ i -> h a b @ i
```

Теорема 18. (Обчислення) Для $z : \mathrm{Join}\ AB$,

$$\begin{cases} \mathrm{JoinRec}(\mathrm{joinl}\ a) \equiv f(a) \\ \mathrm{JoinRec}(\mathrm{joinr}\ b) \equiv g(b) \\ \mathrm{JoinRec}(\mathrm{join}\ a\, b\, @\, i) \equiv h(a, b)\, @\, i \end{cases}$$

Теорема 19. (Унікальність) Будь-які два відображення h_1, h_2 : Join $AB \to C$ є гомотопними, якщо вони збігаються на joinl, joinr і join.

Приклад 5. (З'єднання сфер) З'єднання $S^0 * S^0$ є гомотопічно еквівалентним S^1, оскільки воно з'єднує дві точки (з кожної S^0) шляхами, формуючи структуру, подібну до кола.

2.12 Коліміти

Коліміти конструюють границю послідовності типів, з'єднаних відображеннями, наприклад, пропозиційні відсікання.

Означення 20. (Колiміта) Для послідовності типів $A : \mathrm{nat} \to \mathcal{U}$ і відображень $f : (n : \mathbb{N}) \to An \to A(\mathrm{succ}(n))$, тип коліміти $\mathrm{colimit}(A, f) : \mathcal{U}$.

$$\begin{cases} \mathrm{ix} : (n : \mathrm{nat}) \to An \to \mathrm{colimit}(A, f) \\ \mathrm{gx} : (n : \mathrm{nat}) \to (a : A(n)) \to \mathrm{ix}(\mathrm{succ}(n), f(n, a)) \equiv \mathrm{ix}(n, a) \end{cases}$$

```
def colimit (A : nat -> U) (f : (n : nat) -> A n -> A (succ n)) : U
 := inductive { ix (n : nat) (x: A n)
              | gx (n : nat) (a: A n)
                <i> [ (i=0) -> ix (succ n) (f n a),
                      (i=1) -> ix n a ]
              }
```

Теорема 20. (Елімінація colimit) Для типу P : colimit $Af \to \mathcal{U}$, з $p : (n : \mathrm{nat}) \to (x : An) \to P(\mathrm{ix}(n, x))$ і $q : (n : \mathrm{nat}) \to (a : An) \to \mathrm{PathP}(\langle i \rangle P(\mathrm{gx}(n, a)@i))(p(\mathrm{succ}\ n)(fna))(pna)$, існує $i : \Pi_{x:\mathrm{colimit}\ Af} P(x)$, таке що $i(\mathrm{ix}(n, x)) = pnx$.

2.13 Еквалайзер

Еквалайзер двох відображень $f, g : A \to B$ — це вищий індуктивний тип (HIT), який конструює тип, що складається з елементів у B, де f і g збігаються, разом із шляхами, що забезпечують цю рівність. Це фундаментальна конструкція в теорії гомотопій, яка захоплює підпростір B, де $f(a) = g(a)$ для $a : A$.

Означення 21. (Формування) Для типів $A, B : \mathcal{U}$ і відображень $f, g : A \to B$, еквалайзер coeq $ABfg : \mathcal{U}$.

Означення 22. (Введення) Еквалайзер генерується такою вищою індуктивною композиційною структурою:

$$\begin{cases} \mathrm{inC} : B \to \mathrm{coeq}\ ABfg \\ \mathrm{glueC} : (a : A) \to \mathrm{Path}_{\mathrm{coeq}\ ABfg}(\mathrm{inC}\ (fa), \mathrm{inC}\ (ga)) \end{cases}$$

```
data coeq (A B: U) (f g: A -> B)
   = inC (_: B)
   | glueC (a: A) <i> [(i=0) -> inC (f a), (i=1) -> inC (g a) ]
```

Теорема 21. (Елімінація) Для типу $C : \mathcal{U}$, відображення $h : B \to C$, і сімейства шляхів $y : (x : A) \to \mathrm{Path}_C(h(fx), h(gx))$, існує відображення $\mathrm{coequRec} : \mathrm{coeq}\ ABfg \to C$, таке що:

$$\begin{cases} \mathrm{coequRec}(\mathrm{inC}\ x) = h(x) \\ \mathrm{coequRec}(\mathrm{glueC}\ x \,@\, i) = y(x) \,@\, i \end{cases}$$

```
coequRec (A B C : U) (f g : A -> B) (h: B -> C) (y: (x : A) -> Path C (h (f x)) (h (g x)))
  : (z : coeq A B f g) -> C
  = split
    inC x -> h x
    glueC x @ i -> y x @ i
```

Теорема 22. (Обчислення) Для $z : \mathrm{coeq}\ ABfg$,

$$\begin{cases} \mathrm{coequRec}(\mathrm{inC}\ x) \equiv h(x) \\ \mathrm{coequRec}(\mathrm{glueC}\ x \,@\, i) \equiv y(x) \,@\, i \end{cases}$$

Теорема 23. (Унікальність) Будь-які два відображення $h_1, h_2 : \mathrm{coeq}\ ABfg \to C$ є гомотопними, якщо вони збігаються на inC і glueC, тобто, якщо $h_1(\mathrm{inC}\ x) = h_2(\mathrm{inC}\ x)$ для всіх $x : B$ і $h_1(\mathrm{glueC}\ a) = h_2(\mathrm{glueC}\ a)$ для всіх $a : A$.

Приклад 6. (Еквалайзер як підпростір) Еквалайзер coeq $ABfg$ представляє підпростір B, де $f(a) = g(a)$. Наприклад, якщо $A = B = \mathbb{R}$ і $f(x) = x^2$, $g(x) = x$, еквалайзер захоплює точки, де $x^2 = x$, тобто $\{0, 1\}$.

2.14 Еквалайзер шляхів

Еквалайзер шляхів — це вищий індуктивний тип, який узагальнює еквалайзер для роботи з парами шляхів у B. Дано відображення $p : A \to (b_1, b_2 : B) \times (\mathrm{Path}_B(b_1, b_2)) \times (\mathrm{Path}_B(b_1, b_2))$, він конструює тип, де елементи A породжують пари шляхів між точками в B, із шляхами, що з'єднують кінцеві точки цих шляхів.

Означення 23. (Формування) Для типів $A, B : \mathcal{U}$ і відображення $p : A \to (b_1, b_2 : B) \times (\mathrm{Path}_B(b_1, b_2)) \times (\mathrm{Path}_B(b_1, b_2))$, шляховий еквалайзер $\mathrm{coeqP}\ ABp : \mathcal{U}$.

Означення 24. (Введення) Еквалайзер шляхів генерується такою вищою індуктивною композиційною структурою:

$$\begin{cases} \mathrm{inP} : B \to \mathrm{coeqP}\ ABp \\ \mathrm{glueP} : (a : A) \to \mathrm{Path}_{\mathrm{coeqP}\ ABp}(\mathrm{inP}\ (((p\,a).2.2.1) \,@\, 0), \mathrm{inP}\ (((p\,a).2.2.2) \,@\, 1)) \end{cases}$$

```
data coeqP (A B: U) (p : A -> (b1 b2: B) * (_: Path B b1 b2) * (Path B b1 b2))
  = inP (b: B)
  | glueP (a:A) <i> [(i=0) -> inP (((p a).2.2.1) @ 0), (i=1) -> inP (((p a).2.2.2) @ 1) ]
```

Теорема 24. (Елімінація) Для типу $C : \mathcal{U}$, відображення $h : B \to C$, і сімейства шляхів $y : (a : A) \to \mathrm{Path}_C(h(((p\,a).2.2.1) @ 0), h(((p\,a).2.2.2) @ 1))$, існує відображення coequPRec : coeqP $ABp \to C$, таке що:

$$\begin{cases} \mathrm{coequPRec}(\mathrm{inP}\ b) = h(b) \\ \mathrm{coequPRec}(\mathrm{glueP}\ a @ i) = y(a) @ i \end{cases}$$

```
coequPRec (A B C : U) (p : A -> (b1 b2: B) * (_: Path B b1 b2) * (Path B b1 b2))
          (h: B -> C) (y: (a : A) -> Path C (h (((p a).2.2.1) @ 0)) (h (((p a).2.2.2) @ 1)))
    : (z : coeqP A B p) -> C
   = split
     inP b -> h b
     glueP a @ i -> y a @ i
```

Теорема 25. (Обчислення) Для $z :$ coeqP ABp,

$$\begin{cases} \mathrm{coequPRec}(\mathrm{inP}\ b) \equiv h(b) \\ \mathrm{coequPRec}(\mathrm{glueP}\ a @ i) \equiv y(a) @ i \end{cases}$$

Теорема 26. (Унікальність) Будь-які два відображення $h_1, h_2 :$ coeqP $ABp \to C$ є гомотопними, якщо вони збігаються на inP і glueP, тобто, якщо $h_1(\mathrm{inP}\ b) = h_2(\mathrm{inP}\ b)$ для всіх $b : B$ і $h_1(\mathrm{glueP}\ a) = h_2(\mathrm{glueP}\ a)$ для всіх $a : A$.

Приклад 7. (Шляховий еквалайзер для гомотопії) Шляховий еквалайзер може моделювати простори, де елементи A задають пари шляхів між точками в B. Наприклад, якщо $p(a)$ надає два шляхи від b_1 до b_2 у B, coeqP конструює тип, що з'єднує початкові та кінцеві точки цих шляхів, корисний для вивчення гомотопічних класів.

2.15 K(G,n)

Простори Ейленберга-МакЛейна $K(G,n)$ мають єдину нетривіальну гомотопічну групу $\pi_n(K(G,n)) = G$. Вони визначаються за допомогою відсікань і суспензій.

Означення 25. (K(G,n)) Для абелевої групи G : abgroup, тип $KGnG$: nat $\to \mathcal{U}$.

$$\begin{cases} n = 0 : \mathrm{discreteTopology}(G) \\ n \geq 1 : \mathrm{succ}(n) = \mathrm{nTrunc}(\mathrm{suspension}(K1'(G.1, G.2.1))n)(\mathrm{succ} n) \end{cases}$$

```
KGn (G: abgroup)
   : nat -> U
  = split
    zero -> discreteTopology G
    succ n -> nTrunc (suspension (K1' (G.1,G.2.1)) n) (succ n)
```

Теорема 27. (Елімінація KGn) Для $n \geq 1$, типу $B : \mathcal{U}$ з isNGroupoid$(B, \mathrm{succ}\ n)$, і відображення $f : \mathrm{suspension}(K1'G) \to B$, існує $\mathrm{rec}_{KGn} : KGnG(\mathrm{succ}\ n) \to B$, визначене через nTruncRec.

2.16 Локалізація

Локалізація конструює F-локальний тип із типу X, щодо сімейства відображень $F_A : S(a) \to T(a)$.

Означення 26. (Модальність локалізації) Для сімейства відображень $F_A : S(a) \to T(a)$, F-локалізація $L_F^{AST}(X) : \mathcal{U}$.

$$\begin{cases} \text{center} : X \to L_{F_A}(X) \\ \text{ext} : (a : A) \to (S(a) \to L_{F_A}(X)) \to T(a) \to L_{F_A}(X) \\ \text{isExt} : (a : A) \to (f : S(a) \to L_{F_A}(X)) \to (s : S(a)) \to \text{Path}_{L_{F_A}(X)}(\text{ext } af(Fas), fs) \\ \text{extEq} : (a : A) \to (g, h : T(a) \to L_{F_A}(X)) \to (p : (s : S(a)) \to \text{Path}_{L_{F_A}(X)}(g(Fas), h(Fas))) \to (t : T(a)) \to \\ \text{isExtEq} : (a : A) \to (g, h : T(a) \to L_{F_A}(X)) \to (p : (s : S(a)) \to \text{Path}_{L_{F_A}(X)}(g(Fas), h(Fas))) \to (s : S(a)) \to \end{cases}$$

```
data Localize (A X: U) (S T: A -> U) (F : (x:A) -> S x -> T x)
   = center (x: X)
   | ext (a: A) (f: S a -> Localize A X S T F) (t: T a)
   | isExt (a: A) (f: S a -> Localize A X S T F) (s: S a) <i>
     [ (i=0) -> ext a f (F a s) , (i=1) -> f s ]
   | extEq (a: A) (g h: T a -> Localize A X S T F)
     (p: (s : S a) -> Path (Localize A X S T F) (g (F a s)) (h (F a s)))
     (t : T a) <i> [ (i=0) -> g t , (i=1) -> h t ]
   | isExtEq (a: A) (g h : T a -> Localize A X S T F)
     (p: (s : S a) -> Path (T a -> Localize A X S T F) (g (F a s)) (h (F a s)))
     (s : S a) <i> [ (i=0) -> extEq a g h p (F a s) , (i=1) -> p s ]
```

Теорема 28. (Індукція локалізації) Для будь-якого $P : \Pi_{X:U} L_{F_A}(X) \to U$ з $\{n, r, s\}$, що задовольняють умови когерентності, існує $i : \Pi_{x:L_{F_A}(X)} P(x)$, таке що $i \cdot \text{center}_X = n$.

3 Висновок

HITs кодують CW-комплекси в HoTT, поєднуючи топологію і теорію типів. Вони захоплюють приєднання клітин, з прикладами, такими як сфери, тори та відсікання. Конструктори нескінченності розширюють HITs до нескінченних просторів, оброблювані кубічними примітивами Agda Cubical та індексованими HITs.

Література

[1] The Univalent Foundations Program, *Homotopy Type Theory: Univalent Foundations of Mathematics*, IAS, 2013.

[2] C. Cohen, T. Coquand, S. Huber, A. Mörtberg, *Cubical Type Theory*, Journal of Automated Reasoning, 2018.

[3] A. Mörtberg et al., *Agda Cubical Library*, `https://github.com/agda/cubical`, 2023.

[4] M. Shulman, *Higher Inductive Types in HoTT*, `https://arxiv.org/abs/1705.07088`, 2017.

[5] J. D. Christensen, M. Opie, E. Rijke, L. Scoccola, *Localization in Homotopy Type Theory*, `https://arxiv.org/pdf/1807.04155.pdf`, 2018.

[6] E. Rijke, M. Shulman, B. Spitters, *Modalities in Homotopy Type Theory*, `https://arxiv.org/pdf/1706.07526v6.pdf`, 2017.

[7] M. Riley, E. Finster, D. R. Licata, *Synthetic Spectra via a Monadic and Comonadic Modality*, `https://arxiv.org/pdf/2102.04099.pdf`, 2021.